TRAFFIC NOISE LEVEL PREDICTOR
COMPUTER PROGRAM

MICHIGAN HIGHWAY RESEARCH LABORATORY SECTION

MICHIGAN DEPARTMENT OF STATE HIGHWAYS

TRAFFIC NOISE LEVEL PREDICTOR COMPUTER PROGRAM

G. H. Grove

Research Laboratory Section
Testing and Research Division
Research Project 72 G-189
Research Report No. R-890

Lansing, October 1973

Research Report No. R-890
Traffic Noise Level Predictor Computer Program
Errata Sheet

Page Number	Changes
2	"H1" and "H2" should both read as "HE". The following line numbers should read:

13
14

```
21404 - /"* NCHRP 117. PRØGRAM VERSIØN NØ. 6, 4/1/74.   *"/
21537  111 IF(J.EQ.1.AND.ICØN.EQ.2)DC=DC+DNX;IF(BL.NE.0)GØ TØ 48
21574  D3=DE-DB; D4=DB; IF(HE)201,202,206
21606  205 IF(BL-2.)33,38,38
```

The following lines should be added:

14

```
21575  202 H3=-H;H4=H-HØ;H5=-HØ
21579  GØ TØ 205
21580  201 H3=HE-H;H4=H-HØ;H5=HE-HØ
21581  GØ TØ 204
21582  206 H3=-H;H4=HE-HØ+H;H5=HE-HØ
21583  204 CALL DELS(D2,DEL,DEL6)
21584  H3=H3+8.;H5=H5+8.
21585  CALL DELS(D2,DEL,DL6)
```

Page Number	Changes
19	The third paragraph of Example 1 should have the additional sentence added: "The far lane group traffic data consists of free flowing 4200 vehicles/hour with a 9 percent truck mix and 50 mph truck and 55 mph car speeds."
22, 25, 27	The Parameter Description for DEL3 should read: "Grade correction (0, 2, 3, 5 for $\leq$2, 3-4, 5-6, $\geq$ 7% resp.)."

ABSTRACT

This report contains an update of the original State of Michigan computer program of the NCHRP Report 117 noise prediction model. The original program was distributed to all of the State Highway Departments by the Federal Highway Administration (FHWA) in 1972.

This method for predicting L_{10} (dbA) noise levels due to highway sources was approved by the FHWA in PPM 90-2, "Noise Standards and Procedures," effective January 29, 1973.

Input parameter definitions, flow charts, a listing, and example problems have been included for the user.

TABLE OF CONTENTS

Introduction

Knowledge of the average traffic noise level is not, in itself, necessarily sufficient if one is to define environmental noise acceptability. Some knowledge of the noise peaks and distribution is also required. Although several concepts have been proposed for characterizing these peaks and the distribution, it has been decided that the noisier aspects of the traffic environment can be adequately defined using the temporal unit, L_{10}. This is the noise level which is exceeded 10 percent of the sample time.

Two methods for determining the L_{10} level have been approved by the Federal Highway Administration in PPM 90-2, "Noise Standards and Procedures," effective January 29, 1973. These approved methods are:

1) National Cooperative Highway Research Program Report 117, "Highway Noise: A Design Guide for Highway Engineers," 1971.

2) DOT Transportation Systems Center Report DOT-TSC-FHWA-72-1, "Manual for Highway Noise Prediction," March, 1972.

The Michigan Department of State Highways and Transportation's Research Laboratory has updated its computer program of the so-called "Complete Method" of the above approved method (No. 1).

The program was written with a view towards ease of use on a time-share computer terminal. It allows the user to rapidly determine L_{50} and L_{10} noise levels at any specified distances from the highway for any combinations of the design options available--pavement elevation variables, barrier variables, surface types, grades, etc.[1]

The changes to the original program are the results of:

a. Bolt Beranek and Newman's Report 2209 (1)
b. Bolt Beranek and Newman's Report 2209R (2)
c. Refinements by the author
d. Comments and suggestions from various users in the State Highway Departments and consulting firms across the United States and Canada, and
e. FHWA's "Fundamentals and Abatement of Highway Traffic Noise" Seminar (3), attended by the author August 20-24, 1973 in Chicago, Illinois.

---

[1] Example problems 1 and 2 of Appendix B have 8.2 and 4.9 second execution times, respectively.



* Observer height is measured from the reference plane: position (+) above the plane and negative (-) below the plane.

Figure 1. Pictorial Description of Geometric Parameters.

The program (complete listing in Appendix A) represents the entire "Complete Method" with one exception--it will handle one to eight lanes per lane group. Other choices of "Number of lanes" greater than eight can be achieved by program modification, if desired.

Prospective users are strongly urged to carefully study references (2, 3, 4) prior to using the computer program. These reports present all of the practical and theoretical considerations required to acquaint newcomers to traffic noise prediction techniques.

Input Parameters

The input parameters required are described in abbreviated form on the standard data sheet (Figs. B2 and B5 in Appendix B, a blank data sheet has also been included for users). Geometric parameters are depicted in Figure 1 for a variety of cases.

The program has been set up such that only the data required for a given site geometry will be requested; the data sheet being set up with the proper order of the input variables.

Output of Results

The program prints out intermediate results such as the L_{10} for automobiles (L10A), the L_{10} for trucks (L10T) and a total L_{10} for each lane group of each roadway element.

An overall L_{50} and L_{10} noise level prediction resulting from all roadway elements is then printed along with the distance (DN) to the last element comprising the site. These noise levels are rounded to the nearest dbA.

Two sample problems are given in Appendix B to acquaint the user with expected program operation. The input data and site configurations have been designed to test most of the program features and may not represent a true to life situation. Prospective users can use the results to insure proper operations on their respective computer systems.

In certain situations, the user may need to know the total noise level at a certain point from a multitude of noise sources for which he already has individual results. The decible addition curve of Appendix C will readily accomplish this task.

Program Options and Terminations

Regular Options

After the noise level prediction has been made at the given site for the given traffic conditions, the user will encounter the following message: "INSERT 2, 1, 0, -1 FOR CONTINUE, ITERATE, NEW, STOP," upon which the terminal user will input one of the four suggested numbers. This indicator variable, called ICON, in the actual FORTRAN program is interpreted as follows:

a) ICON = 2 implies that the present site geometry and traffic are the same as the previous calculation but the observer is either closer or further from the roadway. All required distances and angles are automatically adjusted within the program for different observer distances.

b) ICON = 1 implies that we are interested in determining the distance from the roadway (DN) at which a desired L_{10} dbA level exists. All required distances and angles are incremented as a function of the iteration. Presently, this can be used only for single roadway element (NRE = 1) and one or two lane group (N = 1 or 2) geometries. Further expansion of this capability will be completed as time permits.

c) ICON = 0 implies that a noise prediction is desired at a site of different geometry or at the same site with different traffic conditions from the previous prediction.

d) ICON = -1 implies that all desired noise predictions have been completed and the program will terminate.

Default Terminations

a) In the iterate mode (ICON = 1), if the desired L_{10} noise level is selected such as to cause the first incremental step to result in a negative distance for DN, DB, DC, or DS, the following message will appear and the program will default back to the start of a new site location (ICON = 0): "TRY CLOSER L_{10} AFTER RESET OF SITE DATA."

b) Also, if one tries to iterate at a site where there is more than one roadway element (N > 1) or where there is more than two lane groups per roadway element (NRE > 2), the program will terminate with the following message: "SITE TOO COMPLICATED TO ITERATE."

c) In the subroutine ANGLE, if an angle (ANG) greater than 180° is input through an argument in the call statement, the program will default with the following message: "INPUT ANGLE PHYSICALLY IMPOSSIBLE, CHECK DATA."

REFERENCES

1) Kugler, B. A., Piersol, A. G., "Highway Noise: A Field Evaluation of Traffic Noise Reduction Measures," Bolt Beranek and Newman Inc., Report 2209, March 1972.

2) Kugler, B. A., Piersol, A. G., "Highway Noise: A Field Evaluation of Traffic Noise Reduction Measures," Bolt Beranek and Newman Inc., Report 2209R, March 1973 (to be published as NCHRP Report 144).

3) Anderson, G. S., Miller, L. N., Shadley, J. R., "Fundamentals and Abatement of Highway Traffic Noise," Bolt Beranek and Newman Inc., DOT-FH-11-7976, May 1973. (FHWA Region V Seminar held August 20-24, 1973 in Chicago, Illinois.)

4) Gordon, C. G., Galloway, W. J., Kugler, B. A., Nelson, D. L., "Highway Noise: A Design Guide for Highway Engineers," NCHRP Report No. 117, (1971).

APPENDIX A

NOISE PREDICTION COMPUTER PROGRAM

COMPUTER PROGRAM FOR THE NOISE PREDICTION METHOD OF NCHRP REPORT NO. 117 AND 144

Flow charts of the mainline program (Fig. A1) and the five subprograms (Figs. A2 through A4) and the entire program listing are included in this appendix.

Assuming that time share capability is not available to all users, the program can be modified to run on batch processing.

The file statement at the beginning will need to be changed to the proper card reader and printer numbers for the user's system. The WRITE and FORMAT statements which request the insertion of certain traffic and geometric data can be deleted for batch processing. Care should be taken when deleting WRITE statements having line numbers that are addressed from other parts of the program. Multiple statements per card or line exist such as a FORMAT followed by a READ, thus only the FORMAT portion should be removed. The free field format (/) on the READ statements must be modified. Changes involve the following:

FILE	WRITE	FORMAT	READ
21350	21417,21753	21418,21754	21420,21755
	21428,21788	21430,21790	21432,21792
	21434,21800	21436,21801	21439,21801
	21441,21802	21442,21803	21442,21803
	21445,21853	21446,21854	21448,21854
	21468	21470	21470
	21512	21513	21513
	21517	21518	21520
	21532	21534	21536
	21568	21570	21571
	21608	21610	21610

Since this was written for a Burroughs B5500, other CPU's will require the usual changes to cover the few differences in the many versions of FORTRAN IV that exist.

Inquiries regarding this program should be directed to:

Michigan Department of State Highways and Transportation
Highway Research Laboratory
735 East Saginaw Street
Lansing, Michigan 48906 Telephone: 517-373-2730



Figure A1. Mainline flow diagram for calculating L levels by the method of NCHRP Reports 117 and 144.



Figure A2. Flow diagrams for subprograms FIGB10, DBSUM and ANGLE. At left is the "Table Look-up" function, above right the "db Sum" function and below right the "angle increment" subroutine.





Figure A3. Flow diagram for subprogram DELS which is an elevation and shielding correction subroutine.

Figure A4. Flow diagram for subprogram FINBR, which performs the finite barrier length correction.

```
FILE  2=012OUT,UNIT=REMOTE
      COMMON/BLK1/H3,H4,H5,D3,D4,DE
      COMMON/BLK3/RL,ALPHA,THETA
      DIMENSION VAL(12),ARG(12),A1(5),V12(5),V14(5),V18(5)
      DIMENSION V11(5),V13(5),V16(5)
      DIMENSION A5(7),V5(7),A6(7),V6(7),D2(7),DEL(7)
      DIMENSION A2(10),V25(10),V210(10),V215(10)
      DIMENSION Q(4),TMIX(4),ST(4),SA(4),VA(4),VT(4)
      REAL MED
      DATA (VAL(I),I=1,12)/13.1,12.8,12.,10.87,8.19,5.63,
     -4.,3.,2.13,1.5,1.26,1.13/
      DATA (ARG(I),I=1,12)/20.,100.,200.,300.,600.,1500.,3000.,6000.,
     -15000.,40000.,100000.,800000./
      DATA (A1(I),I=1,5)/30.,100.,300.,1000.,3000./
      DATA (V11(I),I=1,5)/8.,0.,-7.,-15.,-22./
      DATA (V12(I),I=1,5)/6.5,-.5,-7.,-15.,-22./
      DATA (V13(I),I=1,5)/6.,-.7,-7.,-15.,-22./
      DATA (V14(I),I=1,5)/5.5,-1.,-7.,-15.,-22./
      DATA (V16(I),I=1,5)/4.,-1.5,-7.,-15.,-22./
      DATA (V18(I),I=1,5)/2.5,-2.,-7.5,-15.,-22./
      DATA (A5(I),I=1,7)/-60.,-20.,20.,40.,60.,70.,80./
      DATA (V5(I),I=1,7)/-.78,-2.03,-4.06,-5.62,-7.82,-9.55,-12.66/
      DATA (A6(I),I=1,7)/0.,10.,20.,40.,60.,100.,160./
      DATA (V6(I),I=1,7)/-16.25,-12.34,-9.68,-6.56,
     --4.68,-2.34,-.31/
      DATA(A2(I),I=1,10)/.1,.2,.3,.4,.5,.6,.7,.8,.9,1./
      DATA(V25(I),I=1,10)/0.,3*-1.,2*-2.,-3.,2*-4.,-5./
      DATA(V210(I),I=1,10)/0.,2*-1.,-2.,2*-3.,-4.,-6.,-7.,-10./
      DATA (DEL(I),I=1,7)/.01,.03,.1,.3,1.,4.,30./
      DATA(V215(I),I=1,10)/0.,-1.,2*-2.,-3.,-4.,-5.,-7.,-10.,-15./
      DATA (D2(I),I=1,7)/-5.,-5.63,-6.88,-8.28,-10.62,-15.,-15./
      WRITE(2,85)
85    FORMAT("*****  *****  *****  *****  *****  *****",
     -"  *****"/"* METHOD APPROVED IN PPM 90-2,  READ REPORT    *"
     -/"* NCHRP 117, PROGRAM VERSION NO. 5, 9/1/73,    *"/
     -"*****  *****  *****  *****  *****  *****  *****")
C
C  PLEASE REPORT ANY PROBLEMS TO G.H.GROVE
C  AT THE MICHIGAN DEPT OF STATE HIGHWAYS
C  T & R LAB IN LANSING, THANK YOU.
      ICON=0
29    INRE=1;ICONB=0
      C1=.017453;C2=57.29578;C3=.0087265;C4=114.59156
      WRITE(2,91)
91    FORMAT(/"INSERT NRE= # OF ROADWAY ELEMENTS")
      READ(2,/)NRE
93    J=1
      DEL2=0;DEL3=0;DEL4=0;DL4=0;DEL5=0;DEL6=0;DL6=0;DEL7=0
      WRITE(2,13)INRE
13    FORMAT(/"INSERT N= # OF LANE GROUPS FOR ROAD ELEMENT #",I2)
      READ(2,/)N
70    WRITE(2,12)J
12    FORMAT(/"INSERT Q,TMIX,ST,SA,HE,DN,RL,BL,P,ID FOR LANE GROUP ",I2)
      READ(2,/)Q(J),TMIX(J),ST(J),SA(J),HE,DN,RL,BL,P,ID
      IF(ID.EQ.0)GO TO 976
      WRITE(2,975)
975   FORMAT("INSERT DEL3,DEL5,DEL7");READ(2,/)DEL3,DEL5,DEL7
976   IF(J.EQ.1)DN1=DN
      IF(N.NE.2)GO TO 98
      WRITE(2,99)
99    FORMAT("INSERT MED")
      READ(2,/)MED
98    CONTINUE
C  CALC, VEHICLE VOLUMES.
      V=ABS(Q(J))
      VT(J)=V*TMIX(J)*.01
      VA(J)=V-VT(J)
C  DEL2 - ELEMENT CORRECTION.
82    IF(RL.EQ.1)GO TO 42
      IF(J.GT.1.OR.ICON.NE.0)GO TO 160
      WRITE(2,32)
32    FORMAT("INSERT THETA");READ(2,/)THETA
160   IF(RL.GT.2.)GO TO 41
      IF(ICON.EQ.0.AND.J.GT.1.OR.ICON.EQ.2)
     -CALL ANGLE(C2,C1,THETA,DN1,DNX)
      DEL2=FIGB10(V5,A5,THETA,7,1)
      GO TO 42
41    IF(ICON.EQ.0.AND.J.GT.1.OR.ICON.EQ.2)
     -CALL ANGLE(C4,C3,THETA,DN1,DNX)
      DEL2=FIGB10(V6,A6,THETA,7,1)
C  DE - EQUIVALENT LANE DISTANCE CALC.
42    DF=DN+12.*P-12.
      DE=SQRT(DN*DF)
C  DEL1 - DISTANCE CORRECTION.
```

```
      IF(P-2.)190,191,192
190   DEL1=FIGB10(V11,A1,DN,5,0);GO TO 61
191   DEL1=FIGB10(V12,A1,DN,5,0);GO TO 61
192   IF(P-4.)193,194,195
193   DEL1=FIGB10(V13,A1,DN,5,0);GO TO 61
194   DEL1=FIGB10(V14,A1,DN,5,0); GO TO 61
195   IF(P.GT.6.)GO TO 196
      DEL1=FIGB10(V16,A1,DN,5,0);GO TO 61
196   DEL1=FIGB10(V18,A1,DN,5,0)
C  DEL4 - VERTICAL CORRECTION.
61    IF(HE.EQ.0.AND.BL.EQ.0.OR.J.NE.1.OR.ICON.NE.0) GO TO 120
      WRITE(2,119)
119   FORMAT("INSERT HO");READ(2,/)HO
120   IF(HE)44,48,53
53    IF(J.GT.1.OR.ICON.NE.0) GO TO 110
      WRITE(2,14)
14    FORMAT("INSERT DS")
      READ(2,/)DS
110   IF(J.EQ.1.AND.ICON.EQ.2)DS=DS+DNX
      H3=0.;H4=HE-HO;H5=H4;D3=DE-DS;D4=DS
      CALL DELS(D2,DEL,DEL4)
      H3=8.;H5=H5+8.
      CALL DELS(D2,DEL,DL4)
      GO TO 48
44    IF(J.GT.1.OR.ICON.NE.0) GO TO 111
      WRITE(2,11)
11    FORMAT("INSERT DC")
      READ(2,/)DC
111   IF(J.EQ.1.AND.ICON.EQ.2)DC=DC+DNX
      H3=HE;H4=-HO;H5=-HO+HE;D3=DE-DC;D4=DC
      CALL DELS(D2,DEL,DEL4)
      H3=H3+8.;H5=H5+8.
      CALL DELS(D7,DEL,DL4)
C  DEL6 - BARRIER CORRECTION.
48    IF(BL.EQ.0.)GO TO 33
      IF(J.GT.1.OR.ICON.NE.0) GO TO 112
      WRITE(2,39)
39    FORMAT("INSERT H,DB")
      READ(2,/)H,DB
      DBX=DB
112   IF(J.EQ.1.AND.ICON.EQ.2)DB=DB+DNX
      H3=-H;H4=H-HO;H5=-HO;D3=DE-DB;D4=DB
      CALL DELS(D2,DEL,DEL6)
      H3=H3+8.;H5=H5+8.
      CALL DELS(D2,DEL,DL6)
      IF(BL-2.)33,38,38
38    IF(J.GT.1.OR.ICON.NE.0) GO TO 113
      WRITE(2,73)
73    FORMAT("INSERT ALPHA");READ(2,/)ALPHA
113   IF(ICON.EQ.2)CALL ANGLE(C4,C3,ALPHA,DB,URX)
      CALL FINBR(V25,V210,V215,A2,DEL6)
      CALL FINBR(V25,V210,V215,A2,DL6)
C  CALC. L50 & L10.
33    S=DEL1+DEL2+DEL7
      SDEL=S+AMAX1(DEL4+DEL6,-20.)+DEL5
      SDELT=S+DEL3+AMAX1(DL4+DL6,-20.)
      YA=.119*VA(J)/SA(J)
      UA=VA(J)*SA(J)*SA(J)*TANH(YA)
      AL50A=10.*ALOG10(UA)-1.+SDEL
      AA=VA(J)*DE/SA(J)
      AL10A=FIGB10(VAL,ARG,AA,12,0)
      OL10A=AL50A+AL10A
      YT=.119*VT(J)/ST(J)
      UT=VT(J)*TANH(YT)/ST(J)
      AL50T=10.*ALOG10(UT)+65.+SDELT
      AT=VT(J)*DE/ST(J)
      AL10T=FIGB10(VAL,ARG,AT,12,0)
      OL10T=AL50T+AL10T
      IF(Q(J).GT.0.)GO TO 51
      OL10A=OL10A+2.
      OL10T=OL10T+4.
51    AL50=DBSUM(AL50A,AL50T)
      AL10=DBSUM(OL10A,OL10T)
      IF(ICON.EQ.1)GO TO 974
      WRITE(2,973)OL10A,OL10T,AL10,INRE,J
973   FORMAT("L10A=",F4.0," L10T=",F4.0," L10=",F4.0,
     -" FOR ELEM. #",I2,", LANE GRP. #",I2)
C  CHECK IF ANY MORE LANE GROUPS.
974   IF(N.EQ.1)GO TO 72
      IF(J.EQ.1)GO TO 65
      AL50=DBSUM(AL50,XX)
      AL10=DBSUM(AL10,YY)
      IF(J.EQ.N)GO TO 72
65    XX=AL50;YY=AL10
```

```
      J=J+1
      IF(N.NE.2)GO TO 70
      DN=DN+MED+12.*P;DNX=DN
      IF(ICON.GT.0)GO TO 98
      WRITE(2,333)
333   FORMAT(/"INSERT #2 Q,TMIX,ST,SA")
      READ(2,/)Q(J),TMIX(J),ST(J),SA(J);GO TO 98
C  CHECK IF ANY MORE ROADWAY ELEMENTS.
72    IF(NRE.EQ.1)GO TO 92
      IF(INRE.EQ.1)GO TO 67
      AL50=DBSUM(AL50,RODL5)
      AL10=DBSUM(AL10,RODL1)
      IF(INRE.EQ.NRE)GO TO 92
67    RODL5=AL50;RODL1=AL10
      INRE=INRE+1
      GO TO 93
C  OUTPUT RESULTING L50 & L10 VALUES.
92    IF(ICON.EQ.1)GO TO 161
      WRITE(2,23)AL50,AL10,NRE,DN1
23    FORMAT(/"*********************"/"L50=",F4.0,"  L10=",
     -F4.0,"  DN1(TO ELEMENT #",I2,")=",F6.0/"*********************"/)
C  CHECK IF ANY MORE PROBLEMS TO BE SOLVED.
      WRITE(2,26)
26    FORMAT("INSERT 2,1,0,-1 FOR CONTINUE,ITERATE,NEW,STOP")
      READ(2,/)ICON
      IF(ICON.GT.1)GO TO 187
      IF(ICON)28,29,30
30    IF(NRE.EQ.1.AND.N.LE.2)GO TO 945
      WRITE(2,940)
940   FORMAT(/"SITE TOO COMPLICATED TO ITERATE");GO TO 28
945   IF(RL.NE.2.OR.BL.NE.2.OR.ICONB.NE.0)GO TO 199
      WRITE(2,185)
185   FORMAT("INSERT BETA");READ(2,/)BETA
199   WRITE(2,141)
141   FORMAT("INSERT DESIRED L10");READ(2,/)AL10D
      DX=0.;DELDN=100.
161   DXN=AL10-AL10D
      IF(ABS(DXN).LT..1)GO TO 162
      IF(DELDN.LT.2.)GO TO 162
      IF(DXN*DX.LT.0.)DELDN=.5*DELDN
      DXX=SIGN(DELDN,DXN)
      DNN=DN1+DXX
      XXX=DNN
      IF(HE)144,145,146
144   DC=DC+DXX
      XXX=DC
      GO TO 145
146   DS=DS+DXX
      XXX=DS
145   IF(BL.EQ.0.)GO TO 181
      DBN=DB+DXX;XXX=DBN
181   IF(XXX.LE.0.)GO TO 180
      IF(RL-2.)151,149,150
150   CALL ANGLE(C4,C3,THETA,DN1,DNN)
151   IF(BL.EQ.2.)CALL ANGLE(C4,C3,ALPHA,DB,DBN)
      GO TO 148
149   CALL ANGLE(C2,C1,THETA,DN1,DNN)
      IF(BL.EQ.2.)GO TO 198
148   DN=DNN;DN1=DNN;DX=DXN
      IF(BL.NE.0.)DB=DBN
171   J=1;INRE=1
      GO TO 82
198   ALB=ALPHA+BETA
      CALL ANGLE(C2,C1,ALB,DN1,DNN)
      CALL ANGLE(C2,C1,BETA,DN1,DNN)
      ALPHA=ALB-BETA
      GO TO 148
162   ICON=0;ICONB=1
      GO TO 92
180   WRITE(2,203)
203   FORMAT("TRY CLOSER L10 AFTER RESET OF SITE DATA")
      ICON=0
      GO TO 29
187   WRITE(2,17)
17    FORMAT("INSERT DN");READ(2,/)DN
      DNX=DN-DN1
      DN1=DN
      GO TO 171
28    STOP
      END
      FUNCTION FIGB10(VAL,ARG,DY,K,J)
      DIMENSION VAL(1),ARG(1)
C  INTERPOLATES ON EITHER A LINEAR OR SEMILOG GRID.
C  IN THE CALL, SET J=1 FOR LINEAR, =0 FOR SEMILOG.
```

```
      D=AMAX1(AMIN1(DY,ARG(K)),ARG(1))
      IF(DY.GE.ARG(1)) GO TO 7
      WRITE(2,20)
20    FORMAT("*** ARGUMENT LIMITED AT LOW END ***")
      WRITE(2,22)DY,ARG(1),ARG(K)
22    FORMAT("DY=",F10.2," ARG(1)=",F6.2," ARG(K)=",F10.2)
      GO TO 8
7     IF(DY.LE.ARG(K)) GO TO 8
      WRITE(2,21)
21    FORMAT("*** ARGUMENT LIMITED AT HIGH END ***")
      WRITE(2,22)DY,ARG(1),ARG(K)
8     DO 1 I=2,K
      IF(D.GT.ARG(I)) GO TO 1
      IF(J.EQ.1) GO TO 9
      FIGB10=ALOG10(D)-ALOG10(ARG(I-1))
      F=ALOG10(ARG(I))-ALOG10(ARG(I-1))
6     FIGB10=FIGB10*(VAL(I)-VAL(I-1))/F+VAL(I-1)
      RETURN
9     FIGB10=D-ARG(I-1)
      F=ARG(I)-ARG(I-1)
      GO TO 6
1     CONTINUE
      RETURN
      END
      FUNCTION DBSUM(A,B)
C   CALCULATES THE DB SUM OF A & B.
      DBSUM=B
      IF(A.LE..01)RETURN
      DBSUM=A
      IF(B.LE..01)RETURN
      DBSUM=10.*ALOG10(10.**(.1*A)+10.**(.1*B))
      RETURN
      END
      SUBROUTINE ANGLE(A,B,ANG,X,XN)
20    FORMAT("INPUT ANGLE PHYSICALLY IMPOSSIBLE,CHECK DATA.")
      IF(ANG+B.LE.1.57077)GO TO 10
      WRITE(2,20);STOP
10    ANG=A+ATAN(TAN(ANG+B)*X/XN)
      RETURN
      END
      SUBROUTINE DELS(D2,DEL,DELX)
C   CALCULATES SHIELDING EFFECTS.
      DIMENSION D2(7),DEL(7)
      COMMON/BLK1/H3,H4,H5,D3,D4,DE
      DELX=0.
      SB=H4/D4;SD=H5/DE
      IF(SD.GT.SB)RETURN
      A=SQRT(H3*H3+D3*D3)
      B=SQRT(H4*H4+D4*D4)
      D=SQRT(H5*H5+DE*DE)
      DL=A+B-D
      IF(DL.LT..01)RETURN
      DELX=FIGB10(D2,DEL,DL,7,0)
      RETURN
      END
      SUBROUTINE FINBR(V25,V210,V215,A2,DELX)
C   CALCULATES FINITE BARRIER ATTENUATION.
      DIMENSION A2(10),V25(10),V210(10),V215(10)
      COMMON/BLK3/RL,ALPHA,THETA
      IF(RL-2.)74,75,76
74    A=ALPHA/180.
      GO TO 77
75    A=ALPHA/(90.-THETA)
      GO TO 77
76    A=ALPHA/THETA
77    IF(A.GE.1.)RETURN
      IF(A.LE..1)GO TO 89
      IF(DELX.GT.-5.)GO TO 86
      IF(DELX.GT.-10.)GO TO 87
      VU=FIGB10(V210,A2,A,10,1)
      VL=FIGB10(V215,A2,A,10,1)
      GO TO 88
89    DELX=0.
      RETURN
86    VU=0.
      VL=FIGB10(V25,A2,A,10,1)
      GO TO 88
87    VU=FIGB10(V25,A2,A,10,1)
      VL=FIGB10(V210,A2,A,10,1)
88    AL=.1*AINT(10.*A)
      AU=AL+.1
      DELX=(A-AL)*(VL-VU)/(AU-AL)+VU
      RETURN
      END
```

APPENDIX B

EXAMPLE PROBLEMS

Example Problem 1

The physical configuration of the roadway selected for this example is such that it is best approximated by a single finite length element with two lane groups each (Fig. B1).

Each lane group consists of four lanes, separated by a 26-ft median, depressed 10 ft with less than 2 percent grade and a rough surface.

The near lane group traffic data consists of free flowing 4,000 vehicles/hour with a 10 percent truck mix and 55 mph truck and 65 mph car speeds.

There is a 100-ft observer distance, 5-ft observer height, 64-ft cut distance and roadway element angle of 50°.

The objective is to find the L_{10} noise level at the observer point for the given geometry and traffic data. If this L_{10} exceeds 70 dbA, then determine the observer distances for the 70 dbA and 60 dbA limits.

The problem would be set up as shown on the data sheet (Fig. B2) and solved as shown on the facsimile of the computer printout (Fig. B3).

The answer was found to be:

L_{10} dbA	Observer distance, ft
75	100
70	122
60	306

Example Problem 2

This example consists of three roadway elements with the following traffic parameters:

Element No. 1: Free flowing 3,300 vehicles/hour, 10 percent truck mix, 50 mph truck and 60 mph car speeds on a non-divided four-lane pavement.

Element No. 2: Interrupted flow of 1,500 vehicles/hour with 11 percent truck mix at 45 mph truck and 50 mph car speeds on the near lane group. Interrupted flow of 1,800 vehicles/hour with 9 percent truck mix at 45 mph truck and 50 mph car speeds on the far lane group.



Figure B1. Roadway geometry for example problem 1.

Element No. 3: Same as for element No. 1 except at 50 mph truck and 55 mph car speeds.

The roadway geometry is illustrated in Figure B4.

The problem being to determine the L_{10} dbA noise level expected at the observer location. This problem can be set up as indicated on the data sheet (Fig. B5) and solved as on the computer printout (Fig. B6) to yield an $L_{10} = 73$ dbA.

DATA SHEET

Date ______

Parameter Description	Symbol	Roadway Element #1		Roadway Element #2		Roadway Element #3	
		Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2
Number of roadway elements	NRE	1					
Number of lane groups per roadway element	N	2					
Hourly flow rate (no sign = free flow, − = interrupted)	Q	4000	4200				
Percent commercial of Q	TMIX	10	9				
Commercial vehicle speed (mph)	ST	55	50				
Automobile speed (mph)	SA	65	55				
Roadway elevation (feet), − = depressed, 0 = at grade, + = elevated	HE	-10					
Observer to center of near lane (feet)	DN	100					
Roadway length type (1 = ∞, 2 = semi-∞, 3 = finite)	RL	3					
Barrier length (0 = no barrier, 1 = ∞, 2 = finite)	BL	0					
Number of lanes per lane group	P	4					
(0 ⇒ DEL 3, 5, 7 = 0, 1 ⇒ one or more ≠ 0)	ID	1					
Grade correction (0, 2, 3, 4 for ⩽ 2, 3-4, 5-6, ⩾ 7% resp.)	DEL3	0					
Roadway surface correction (-5, 0, 5 for smooth, normal, rough)	DEL5	5					
Structure shield correction (-4.5/1st row houses, -1.5/others, -10 max)	DEL7	0					
Median width for divided highways (feet)	MED	26					
Roadway element angle (degrees); only for RL ≠ 1	THETA	50					
Observer height rel. to ref. plane (feet); + above, − below	HO	5					
Observer to shoulder (feet); only for HE > 0	DS						
Observer to cut (feet); only for HE < 0	DC	64					
Barrier height (feet) } only for BL ≠ 0	H						
Observer to barrier } only for BL ≠ 0	DB						
Barrier included angle (degrees); only for BL = 2	ALPHA						
Barrier end-normal angle (degrees); only for RL = 2 and BL = 2	BETA						
	L_{50}		69				
	L_{10}		75				

Figure B2. Data sheet for example problem 1.

```
*****  *****  *****  *****  *****  *****  *****
* METHØD APPRØVED IN PPM 90-2.  READ REPØRT   *
* NCHRP 117. PRØGRAM VERSIØN NØ. 5, 9/1/73.   *
*****  *****  *****  *****  *****  *****  *****

INSERT NRE= # ØF RØADWAY ELEMENTS
?1←

INSERT N= # ØF LANE GRØUPS FØR RØAD ELEMENT # 1
?2←

INSERT Q,TMIX,ST,SA,HE,DN,RL,BL,P,ID FØR LANE GRØUP  1
?4000,10,55,65,-10,100,3,0,4,1←
INSERT DEL3,DEL5,DEL7
?0,5,0←
INSERT MED
?26←
INSERT THETA
?50←
INSERT HØ
?5←
INSERT DC
?64←
L10A= 65. L10T= 73. L10= 73. FØR ELEM. # 1, LANE GRP. # 1

INSERT #2 Q,TMIX,ST,SA
?4200,9,50,55←
L10A= 65. L10T= 66. L10= 68. FØR ELEM. # 1, LANE GRP. # 2

********************
L50= 69.  L10= 75.  DNI(TØ ELEMENT # 1)=  100.
********************

INSERT 2,1,0,-1 FØR CØNTINUE,ITERATE,NEW,STØP
?1←
INSERT DESIRED L10
?70←

********************
L50= 65.  L10= 70.  DNI(TØ ELEMENT # 1)=  122.
********************

INSERT 2,1,0,-1 FØR CØNTINUE,ITERATE,NEW,STØP
?1←
INSERT DESIRED L10
?60←

********************
L50= 56.  L10= 60.  DNI(TØ ELEMENT # 1)=  306.
********************

INSERT 2,1,0,-1 FØR CØNTINUE,ITERATE,NEW,STØP
?-1←
```

Figure B3. Facsimile of computer printout for example problem 1.



Figure B4. Roadway geometry for example problem 2.

DATA SHEET

Date ________

Parameter Description	Symbol	Roadway Element #1		Roadway Element #2		Roadway Element #3	
		Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2
Number of roadway elements	NRE	3					
Number of lane groups per roadway element	N	1		2		1	
Hourly flow rate (no sign = free flow, − = interrupted)	Q	3300		-1500	-1800	3300	
Percent commercial of Q	TMIX	10		11	9	10	
Commercial vehicle speed (mph)	ST	50		45	45	50	
Automobile speed (mph)	SA	60		50	50	55	
Roadway elevation (feet), − = depressed, 0 = at grade, + = elevated	HE	-5		0		12	
Observer to center of near lane (feet)	DN	200		220		400	
Roadway length type (1 = ∞, 2 = semi-∞, 3 = finite)	RL	3		3		2	
Barrier length (0 = no barrier, 1 = ∞, 2 = finite)	BL	0		2		0	
Number of lanes per lane group	P	4		2		4	
(0 ⇒ DEL 3, 5, 7 = 0, 1 ⇒ one or more ≠ 0)	ID	1		1		0	
Grade correction (0, 2, 3, 4 for ⩽ 2, 3-4, 5-6, ⩾ 7% resp.)	DEL3	2		2			
Roadway surface correction (-5, 0, 5 for smooth, normal, rough)	DEL5	-5		5			
Structure shield correction (-4.5/1st row houses, -1.5/others, -10 max)	DEL7	0		-4.5			
Median width for divided highways (feet)	MED			50			
Roadway element angle (degrees); only for RL ≠ 1	THETA	75		88		12	
Observer height rel. to ref. plane (feet); + above, − below	HO	5		5		-7	
Observer to shoulder (feet); only for HE > 0	DS					380	
Observer to cut (feet); only for HE < 0	DC	180					
Barrier height (feet) } only for BL ≠ 0	H			10			
Observer to barrier } only for BL ≠ 0	DB			180			
Barrier included angle (degrees); only for BL = 2	ALPHA			52			
Barrier end-normal angle (degrees); only for RL = 2 and BL = 2	BETA						
	L_{50}						67
	L_{10}						73

Figure B5. Data sheet for example problem 2.

```
INSERT NRE= # ØF RØADWAY ELEMENTS
?3←

INSERT N= # ØF LANE GRØUPS FØR RØAD ELEMENT # 1
?1←

INSERT Q,TMIX,ST,SA,HE,DN,RL,BL,P,ID FØR LANE GRØUP  1
?3300,10,50,60,-5,200,3,0,4,1←
INSERT DEL3,DEL5,DEL7
?2,-5,0←
INSERT THETA
?75←
INSERT HØ
?5←
INSERT DC
?180←
L10A= 51. L10T= 71. L10= 71. FØR ELEM. # 1, LANE GRP. # 1

INSERT N= # ØF LANE GRØUPS FØR RØAD ELEMENT # 2
?2←

INSERT Q,TMIX,ST,SA,HE,DN,RL,BL,P,ID FØR LANE GRØUP  1
?-1500,11,45,50,0,220,3,2,2,1←
INSERT DEL3,DEL5,DEL7
?2,5,-4.5←
INSERT MED
?50←
INSERT THETA
?88←
INSERT HØ
?5←
INSERT H,DB
?10,180←
INSERT ALPHA
?52←
L10A= 58. L10T= 64. L10= 65. FØR ELEM. # 2, LANE GRP. # 1

INSERT #2 Q,TMIX,ST,SA
?-1800,9,45,50←
L10A= 56. L10T= 60. L10= 62. FØR ELEM. # 2, LANE GRP. # 2

INSERT N= # ØF LANE GRØUPS FØR RØAD ELEMENT # 3
?1←

INSERT Q,TMIX,ST,SA,HE,DN,RL,BL,P,ID FØR LANE GRØUP  1
?3300,10,50,55,12,400,2,0,4,0←
INSERT THETA
?12←
INSERT HØ
?-7←
INSERT DS
?380←
L10A= 52. L10T= 63. L10= 63. FØR ELEM. # 3, LANE GRP. # 1

********************
L50= 67.  L10= 73.  DNI(TØ ELEMENT # 3)=  400.
********************

INSERT 2,1,0,-1 FØR CØNTINUE,ITERATE,NEW,STØP
?-1←
```

Figure B6. Facsimile of computer printout for example problem 2.

DATA SHEET

Date ________

Parameter Description	Symbol	Roadway Element #1		Roadway Element #2		Roadway Element #3	
		Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2	Lane Group #1	Lane Group #2
Number of roadway elements	NRE						
Number of lane groups per roadway element	N						
Hourly flow rate (no sign = free flow, - = interrupted)	Q						
Percent commercial of Q	TMIX						
Commercial vehicle speed (mph)	ST						
Automobile speed (mph)	SA						
Roadway elevation (feet), - = depressed, 0 = at grade, + = elevated	HE						
Observer to center of near lane (feet)	DN						
Roadway length type (1 = ∞, 2 = semi-∞, 3 = finite)	RL						
Barrier length (0 = no barrier, 1 = ∞, 2 = finite)	BL						
Number of lanes per lane group	P						
(0 ⇒ DEL 3, 5, 7 = 0, 1 ⇒ one or more ≠ 0)	ID						
Grade correction (0, 2, 3, 4 for ⩽ 2, 3-4, 5-6, ⩾ 7% resp.)	DEL3						
Roadway surface correction (-5, 0, 5 for smooth, normal, rough)	DEL5						
Structure shield correction (-4.5/1st row houses, -1.5/others, -10 max)	DEL7						
Median width for divided highways (feet)	MED						
Roadway element angle (degrees); only for RL ≠ 1	THETA						
Observer height rel. to ref. plane (feet); + above, - below	HO						
Observer to shoulder (feet); only for HE > 0	DS						
Observer to cut (feet); only for HE < 0	DC						
Barrier height (feet) } only for BL ≠ 0	H						
Observer to barrier } only for BL ≠ 0	DB						
Barrier included angle (degrees); only for BL = 2	ALPHA						
Barrier end-normal angle (degrees); only for RL = 2 and BL = 2	BETA						
	L_{50}	____	____	____	____	____	____
	L_{10}	____	____	____	____	____	____

APPENDIX C

ADDITION OF NOISE LEVELS

ADDITION OF DECIBEL LEVELS

The addition of decibel levels is accomplished by use of the graph as follows:

1. Call the larger level A, the smaller B.
2. Find the numerical difference between A and B, i.e., A - B.
3. Enter the graph on the absicissa (x) at A - B. On the ordinate (y) read C.
4. Add the numerical sum of A plus C to obtain the true combined decibel level, D.

EXAMPLE:

1. A = 83 dbA, B = 77 dbA
2. A - B = 6 dbA
3. C = 1 dbA
4. D = A + C = 83 + 1 = 84 dbA



Figure C1. Summing Decibel Levels